order and multiplication of natural numbers

Johan G. F.Belinfante and Claudia D. Huang
2005 June 28

```
In[1]:= SetDirectory["i:"]; << goedel70.27b; << tools.m

:Package Title: goedel70.27b       2005 June 27 at 4:50 p.m.

It is now:  2005 Jun 28 at 22:17

Loading Simplification Rules

TOOLS.M                      Revised 2005 June 17

weightlimit = 40
```

summary

Quaife's **O** group contains a large number of theorems about order properties of natural numbers. The focus of this notebook is on deriving new rewrite rules to enable the **GOEDEL** program to recognize some theorems in this group involving multiplication.

```
In[2]:= "Art Quaife, Automated Development of Fundamental Mathematical Theories,
           Kluwer Academic Publishers, Dordrecht, the Netherlands, 1992.";
```

The rewrite rules about multiplication of natural numbers derived in this noteboook include ones related to Quaife's theorem **(M9)**, the first clause of **(O21)**, and his theorems **(O22)**, **(O23)**, **(O24)** and **(O27)**. Theorem **(M9)** states that multiplication by natural numbers distributes over floored subtraction, here called **monus**. Quaife's first clause of his **(O21)**, and his theorems **(O22)**, **(O23)** and **(O24)** are consequences of a rewrite rule about cancelling a common nonzero factor **x** in an inequality of the form **x y < x z**. The transitive law for order and this cancellation law are used to derive Quaife's theorem **(O27)**, which states that strict inequalities involving natural numbers can be multiplied.

Quaife's (M9)

The following technical lemma is made into a temporary rewrite rule.

```
In[3]:= equal[intersection[complement[image[V, intersection[nat[z], set[nat[y]]]]],
          natsub[natmul[nat[x], nat[y]], natmul[nat[x], nat[z]]]],
         nat[natsub[natmul[nat[x], nat[y]], natmul[nat[x], nat[z]]]]]

Out[3]= True

In[4]:= intersection[complement[image[V, intersection[nat[z_], set[nat[y_]]]]],
          natsub[natmul[nat[x_], nat[y_]], natmul[nat[x_], nat[z_]]]] :=
         nat[natsub[natmul[nat[x], nat[y]], natmul[nat[x], nat[z]]]]
```

The following rewrite rule says that **natmul** distributes over **monus.**

```
In[5]:= Map[complement, Map[complement[APPLY[composite[NATMUL, #], nat[x]]] &,
            RIGHT[intersection[v, complement[image[V, w]]]] // VSNormality]] /.
         {v → natsub[nat[y], nat[z]], w → intersection[nat[z], set[nat[y]]]}

Out[5]= natmul[nat[x], nat[natsub[nat[y], nat[z]]]] ==
         nat[natsub[natmul[nat[x], nat[y]], natmul[nat[x], nat[z]]]]

In[6]:= natmul[nat[x_], nat[natsub[nat[y_], nat[z_]]]] :=
         nat[natsub[natmul[nat[x], nat[y]], natmul[nat[x], nat[z]]]]
```

Restatement of Quaife's **(M9):**

```
In[7]:= equal[monus[natmul[nat[x], nat[y]], natmul[nat[x], nat[z]]],
         natmul[nat[x], monus[nat[y], nat[z]]]]

Out[7]= True
```

---

a cancellation law

The following cancellation law holds:

```
In[8]:= Map[not, SubstTest[subclass, nat[u], nat[v],
           {u → natmul[nat[x], nat[z]], v → natmul[nat[x], nat[y]]}]] // Reverse

Out[8]= member[natmul[nat[x], nat[y]], natmul[nat[x], nat[z]]] ==
         and[member[nat[y], nat[z]], not[equal[0, nat[x]]]]

In[9]:= member[natmul[nat[x_], nat[y_]], natmul[nat[x_], nat[z_]]] :=
         and[member[nat[y], nat[z]], not[equal[0, nat[x]]]]
```

This rewrite rule implies Quaife's first clause of **(O21)** and theorems **(O22)**, **(O23)** and **(O24)**.

a special case

The following special case is of interest:

```
In[10]:= SubstTest[member, natmul[nat[x], nat[z]], natmul[nat[x], nat[y]], z → set[0]]

Out[10]= member[nat[x], natmul[nat[x], nat[y]]] ==
          and[member[set[0], nat[y]], not[equal[0, nat[x]]]]

In[11]:= member[nat[x_], natmul[nat[x_], nat[y_]]] :=
          and[member[set[0], nat[y]], not[equal[0, nat[x]]]]
```

There already exists a rewrite rule for an inequality going in the reverse direction. Combining these two inequalities yields the fact that one is the only nonzero number not greater than one. The same conclusion can also be derived somewhat more directly as follows:

```
In[12]:= SubstTest[and, not[member[nat[x], nat[y]]],
          not[member[nat[y], nat[x]]], y → set[0]]

Out[12]= and[not[equal[0, nat[x]]], not[member[set[0], nat[x]]]] ==
          equal[nat[x], set[0]]

In[13]:= and[not[equal[0, nat[x_]]], not[member[set[0], nat[x_]]]] :=
          equal[nat[x], set[0]]
```

derivation of Quaife's Theorem (O27)

A double application of the cancellation law yields this lemma:

```
In[14]:= SubstTest[implies, and[member[nat[t], nat[w]], member[nat[w], nat[z]]],
          member[nat[t], nat[z]], {t → natmul[nat[u], nat[v]],
           w → natmul[nat[u], nat[y]], z → natmul[nat[x], nat[y]]}]

Out[14]= or[equal[0, nat[u]], equal[0, nat[y]],
           member[natmul[nat[u], nat[v]], natmul[nat[x], nat[y]]],
           not[member[nat[u], nat[x]]], not[member[nat[v], nat[y]]]] == True

In[15]:= (% /. {u → u_, v → v_, x → x_, y → y_}) /. Equal → SetDelayed
```

The two equality literals are redundant and can be removed. The result is Quaife's theorem **(O27)**.

```
In[16]:= Map[not, SubstTest[and, implies[and[p1, p2], or[p3, p4, p5]],
           implies[p2, not[p3]], implies[and[p1, p2, p4], p5],
           not[implies[and[p1, p2], p5]], {p1 → member[nat[u], nat[x]],
            p2 → member[nat[v], nat[y]], p3 → equal[0, nat[y]], p4 → equal[0, nat[u]],
            p5 → member[natmul[nat[u], nat[v]], natmul[nat[x], nat[y]]]}]]

Out[16]= or[member[natmul[nat[u], nat[v]], natmul[nat[x], nat[y]]],
           not[member[nat[u], nat[x]]], not[member[nat[v], nat[y]]]] == True

In[17]:= or[member[natmul[nat[u_], nat[v_]], natmul[nat[x_], nat[y_]]],
           not[member[nat[u_], nat[x_]]], not[member[nat[v_], nat[y_]]]] := True
```